## Friday, July 15, 2016

### Resonance in pseudoscalar Yukawa theory

A post with calculations in pseudoscalar Yukawa theory and plots of cross sections to illustrate a resonance.
The pseudoscalar Yukawa theory has Lagrangian \begin{equation*} \mathcal{L} = \frac{1}{2} \left( \partial_{\mu} \phi \right)^2 - \frac{1}{2} M^2 \phi^2 + \bar \psi \left( i \gamma^{\mu} \partial_{\mu} \psi - m \right) \psi + \frac{\lambda}{4!} \phi^4 + i g \phi \bar \psi \gamma^5 \psi \end{equation*} The physical interpretation of this Lagrangian is as follows. $\phi$ is a real scalar field describing a boson with spin $0$ and mass $M$. $\psi$ is a Dirac spinor describing an electron and a positron with mass $m$. These particles interact with one another by exchanging the boson $\phi$.

Decay rate of boson
If $M > 2 m$, then the boson can decay into an electron and a positron. At tree level, the decay rate is [1] \begin{equation*} \Gamma = \frac{g^2}{8 \pi} \sqrt{M^2 - 4m^2} \end{equation*}
For example, if $g = 1$, $m = 1\ \text{GeV}$ and $M = 4\ \text{GeV}$, then the lifetime of the boson is $4.8\times 10^{-24} s$.

Scattering at tree level
The spin averaged amplitude for the scattering $e^+ e^- \to e^+ e^-$ is [2] $$\label{eq:20160713c} \langle | \mathcal{M} |^2 \rangle = g^4 \left[ \frac{ s^2 }{(M^2 -s)^2} + \frac{s t}{(M^2 - s) (M^2 - t)} + \frac{ t^2}{(M^2 -t)^2} \right]$$ From \eqref{eq:20160713c} one can calculate the total cross section. In the center of mass frame this is \begin{equation*} \sigma = 2 \pi \int_0^{\pi}\!\! d\theta\ \sin\theta \frac{d\sigma}{d\Omega} \qquad\text{with}\qquad \frac{d\sigma}{d\Omega} = \left( \frac{1}{ 8 \pi E} \right)^2 \langle | \mathcal{M} |^2 \rangle \end{equation*} Here $E$ is the center of mass energy, i.e. the sum of the energy of the incoming electron and positron. I plot $\sigma$ as function of $E$. I take again $g = 1$, $m = 1\ \text{GeV}$ and $M = 4\ \text{GeV}$. The Mathematica code can be found at the bottom of this post.

 Total cross section of $e^+ e^- \to e^+ e^-$ in pseudoscalar Yukawa theory
In the graph there is a singularity at $E = M$. For this energy, the incoming electron and positron have enough energy to produce the boson and then decay.

In the rest of the post, I calculate the loop correction to the boson propagator if $M > 2 m$. I decided to do this calculation because Srednicki only calculates the loop correction in the case $M < 2 m$. I find the case $M > 2m$ more interesting, because adding loops to the boson propagator resolves the singularity in the total cross section.

Loop correction to the boson propagator
If $M > 2 m$, the sum of the 1PI diagrams of the boson propagator is [3] \begin{equation*} \Pi(p) = \frac{g^2}{4 \pi^2} \left( \int_0^1\!\!\! dx \left( m^2 - 3 x (1-x) p^2 \right) \log \frac{D(x,p)}{| D(x,M) |} + \kappa\ (M^2 - p^2)\right) \end{equation*} with $$\label{eq:20160715a} \kappa = \int_0^1\!\!\! dx\ x (1-x) \frac{3 x (1-x) M^2 - m^2}{D(x,M)}$$ and \begin{equation*} D(x,p) = - x(1-x) p^2 + m^2 \end{equation*}

Scattering at one loop
The spin averaged amplitude for the scattering $e^+ e^- \to e^+ e^-$, including the loops in the boson propagator, is $$\label{eq:20160713e} \langle | \mathcal{M} |^2 \rangle = g^4 \left[ s^2 | \Delta(p_1 + p_2)|^2 + \frac{1}{2} s t \left( \Delta(p_1 -p_3)\overline{\Delta(p_1 + p_2)} +\Delta(p_1 + p_2)\overline{\Delta(p_1 - p_3)}\right)+ t^2 | \Delta(p_1 - p_3) |^2 \right]$$ with \begin{equation*} \Delta(p) = \frac{1}{p^2 - M^2 + \Pi(p)} \end{equation*} I again plot the cross section for the same values $g = 1$, $m = 1\ \text{GeV}$ and $M = 4\ \text{GeV}$, now using \eqref{eq:20160713e}

 Total cross section of $e^+ e^- \to e^+ e^-$ in pseudoscalar Yukawa theory black line: tree level; red line: with loop correction in boson propagator
One can thus see that the singularity is resolved by the addition of the loops in the boson propagator.

It is also known that the width of the resonance is related to the lifetime of the particle. The shorter the particle lives, the wider the resonance. If $M$ increases, $\Gamma$ increases, thus the lifetime decreases, thus the resonance becomes wider. This is illustrated in the next graph.
 Total cross section of $e^+ e^- \to e^+ e^-$ for $M=4\ \text{GeV}$ and $M = 10\ \text{GeV}$
References
[1] This is exercise 48.4 in Srednicki. The answer can be checked against notes by A. George.

[2] I obtained this formula as follows. Srednicki calculates the amplitude for the same process in the Yukawa theory (not the pseudoscalar Yukawa theory) in chapter 48. I repeated his calculation, changing $g$ to $i g \gamma^5$ at the appropriate places. This leads to extra factors of $\gamma^5$ in the trace formulas as compared to the Yukawa theory, finally giving formula \eqref{eq:20160713c}.

[3] I obtained this formula as follows. Srednicki calculates the loop correction in the case $M < 2 m$ in chapter 51. The case $M > 2 m$ is the same apart from the on-shell conditions. In the case $M < 2m$ one can impose $\Pi(M) = \Pi'(M)=0$. In the case $M > 2 m$ one can only impose $\Re ( \Pi(M) ) = \Re ( \Pi'(M) )=0$. This is explained on page 152 in Srednicki.

[4] As written in \eqref{eq:20160715a}, the integral for $\kappa$ diverges. The integral should be interpreted as $$\label{eq:20160715b} \kappa = \int_0^1\!\!\! dx\ x (1-x) \frac{3 x (1-x) M^2 - m^2 + i \epsilon}{- x (1-x) M^2 + m^2 - i \epsilon}$$ with $\epsilon$ a small positive real number. The integral can then be calculated for non-zero $\epsilon$. After taking $\epsilon$ to zero, the result is \begin{equation*} \kappa = -\frac{1}{2} - \frac{2 m^2}{M^2} + \frac{m^4}{M^4} \frac{1}{W} \log\frac{1- W}{1+W} \end{equation*} with $W = \sqrt{1 - 4 m^2 / M^2}$

Mathematica code for graph at tree level
 dim = 0.389; (* GeV^2 mbarn from http://pdg.lbl.gov/1998/consrpp.pdf*) 

pseudoM2[g_,m_,M_,s_,t_,u_]:= g^4 ( s^2/(s - M^2)^2+ s t /((s - M^2) (t - M^2)) + t^2/(t - M^2)^2 )

pseudod\[Sigma][g_,m_,M_,EE_,\[Theta]_]:= Module[{s = (2 EE)^2, t = - 4 (EE^2 - m^2) Sin[\[Theta]/2]^2, u = - 4 (EE^2 - m^2) Cos[\[Theta]/2]^2},dim /(16 Pi EE)^2 pseudoM2[g,m,M,s,t,u]]

pseudo\[Sigma][g_?NumericQ,m_?NumericQ,M_?NumericQ,EE_?NumericQ]:= 2 Pi NIntegrate[pseudod\[Sigma][g,m,M,EE,\[Theta]] Sin[\[Theta]],{\[Theta],0,Pi}] 

Mathematica code for graph at loop level
 \[CapitalDelta][x_,m_,p2_]:= m^2 - x( 1-x) p2

mylog[z_]/; z \[Element] Reals:= If[z>0,Log[z],If[z<0, Log[Abs[z]] - I Pi]]

I include the singularities in the integration specification to help NIntegrate, see Mathematica help pages

\[CapitalPi][g_?NumericQ,m_?NumericQ,M_?NumericQ,p2_?NumericQ] /;M > 2 m:= Module[{term1, term2, term3,x1,x2,x,W},
x1 = 1/2 ( 1 - Sqrt[1 - 4 m^2/p2]);
x2 = 1/2 ( 1 + Sqrt[1 - 4 m^2/p2]);
term1 = If[p2 < 4 m^2, NIntegrate[ ( m^2 - 3 x (1-x) p2) mylog[\[CapitalDelta][x,m,p2]] ,{x,0,1},PrecisionGoal->3], NIntegrate[ ( m^2 - 3 x (1-x) p2) mylog[\[CapitalDelta][x,m,p2]] ,{x,0,x1,x2,1},PrecisionGoal->3]];
x1 = 1/2 ( 1 - Sqrt[1 - 4 m^2/M^2]);
x2 = 1/2 ( 1 + Sqrt[1 - 4 m^2/M^2]);
term2 = NIntegrate[ ( m^2 - 3 x (1-x) p2) mylog[Abs[\[CapitalDelta][x,m,M^2]]] ,{x,0,x1,x2,1},PrecisionGoal->3];
W = Sqrt[1 - 4 m^2/M^2];
term3 = -(1/2)-2 m^2/ M^2+4 m^4 / (M^4 W) Log[(1 - W)/(1 + W)];
g^2/(4 Pi^2)(term1 - term2 + term3 (M^2 - p2))]

PropLoop[g_,m_,M_,p2_]:= 1 / (p2 - M^2 + \[CapitalPi][g,m,M,p2])

pseudoM2WithLoop[g_,m_,M_,s_,t_,u_]:= g^4 ( s^2 Abs[PropLoop[g,m,M,s]]^2+ 1/2 s t (PropLoop[g,m,M,s] Conjugate [PropLoop[g,m,M,t]] + PropLoop[g,m,M,t] Conjugate [PropLoop[g,m,M,s]]) +  t^2Abs[PropLoop[g,m,M,t]]^2 )

pseudod\[Sigma]WithLoop[g_,m_,M_,EE_,\[Theta]_]:= Module[{s = (2 EE)^2, t = - 4 (EE^2 - m^2) Sin[\[Theta]/2]^2, u = - 4 (EE^2 - m^2) Cos[\[Theta]/2]^2},dim /(16 Pi EE)^2 pseudoM2WithLoop[g,m,M,s,t,u]]

pseudo\[Sigma]WithLoop[g_?NumericQ,m_?NumericQ,M_?NumericQ,EE_?NumericQ]:= 2 Pi NIntegrate[pseudod\[Sigma]WithLoop[g,m,M,EE,\[Theta]] Sin[\[Theta]],{\[Theta],0,Pi},PrecisionGoal->2]